

Question 5 (30 points):

Null-terminated strings are sorted based on a lexicographic comparison of their characters. Given two strings S_a and S_b , the result of their lexicographic comparison can have these outcomes:

- $S_a = S_b$ if $S_a[i] = S_b[i]$ for every character in S_a and in S_b .
- $S_a < S_b$ if $\exists j$ such that $S_a[i] = S_b[i]$ for every $i < j$ and $S_a[j] < S_b[j]$.

Therefore, to lexicographically compare two strings, we compare the ASCII codes of the strings character by character until we find the first position in which the strings are different. Whichever string has a smaller character at that position is the smaller string.

When comparing two strings of different lengths, if one of the strings is a substring of the other, the shorter string is lexicographically smaller. For example the null-terminated string `grade` is lexicographically smaller than the null-terminated string `gradeA`.

In this question you will create two functions to sort a vector of pointers to strings. Lets call this vector `S`. Each element of `S` contains a pointer to a null-terminated string (the address of the first character of a null-terminated string). The end of `S` is signalled by the sentinel value -1.

The task is to sort the pointers in `S` such that the strings are sorted in lexicographic order. To accomplish this task, you will write two functions in RISC-V assembly. The first function, `LexiCmp`, compares two string and returns -1, 0, or 1, depending on the result of the lexicographic comparison of the two strings. The second function, `LexiSort`, sorts the pointers in `S` according to their lexicographic order. `LexiSort` must call `LexiCmp` in order to compare strings.

Both functions must follow all the RISC-V register saving and restoring conventions. The two functions must be written independently, thus no assumptions about how the other function uses registers can be made.

The solution must work for any null-terminated strings, including the empty string. An empty string is lexicographically smaller than any non-empty string.

1. **(10 points)** Write RISC-V assembly code for a function called `LexiCmp`. It has two parameters that are pointers to strings and a single return value.

parameters:

- `a0`: pointer to a string S_a
- `a1`: pointer to a string S_b

return value:

- `a0` = 0 if $S_a = S_b$
- `a0` = 1 if $S_a > S_b$
- `a0` = -1 if $S_a < S_b$

2. **(20 points)** Write `LexiSort`, which implements the Bubble Sort algorithm to sort the pointers to strings in `S`. The single parameter of `LexiSort` is the address of the first string pointer in `S`. `LexiSort` has no return values. After the execution of `LexiSort` the pointers in `S` are sorted according to the lexicographic order of the strings.

```

2  # LexiCmp
3  # arguments: a0: pointer to string Sa
4  #             a1: pointer to string Sb
5  # return value: a0: -1 if Sa<Sb; 1 if Sa>Sb; 0 if Sa=Sb
6  #
7  LexiCmp:
8  NextChar:
9      lb      t0, 0(a0)          # t0 <- *Sa
10     lb      t1, 0(a1)          # t1 <- *Sb
11     blt     t0, t1, LessThan    # if t0<t1 then Sa<Sb
12     bgt     t0, t1, GreaterThan # if t0>t1 then Sa>Sb
13     addi    a0, a0, 1          # Sa++
14     addi    a1, a1, 1          # Sb++
15     bne     t0, zero, NextChar  # Not end of string
16     mv      a0, zero           # Sa=Sb thus a0 <- 0
17     jalr    zero, ra, 0
18 LessThan:
19     addi    a0, zero, -1        # a0 <- -1
20     jalr    zero, ra, 0
21 GreaterThan:
22     addi    a0, zero, 1         # a0 <- 1
23     jalr    zero, ra, 0

```

Figure 1: A solution for LexiCmp.

<pre> 28 void LexiSort(char **S){ 29 int i, j; 30 int Ctemp; 31 char *Stempj, *Stempj1; 32 for (i = 0; S[i] != -1; i += 1){ 33 for (j = i-1; j >= 0; j -= 1){ 34 Stempj = S[j]; 35 Stempj1 = S[j+1]; 36 Ctemp = LexiCmp(Stempj, Stempj1); 37 if(Ctemp <= 0){ 38 break; 39 } 40 S[j] = Stempj1; 41 S[j+1] = Stempj; 42 } 43 } 44 } </pre>	<pre> 95 void LexiSort(char **S){ 96 int Ctemp; 97 char *Rtemp, *Rtemp1; 98 char **T, **R; 99 for (T = S ; *T != -1; T++){ 100 R = T; 101 for (R--; R >= S; R--){ 102 Rtemp = *R; 103 Rtemp1 = *(R+1) 104 Ctemp = LexiCmp(Rtemp, Rtemp1); 105 if(Ctemp <= 0){ 106 break; 107 } 108 *R = Rtemp1; 109 *(R+1) = Rtemp; 110 } 111 } 112 } </pre>
--	--

(a) Index-based Sorting

(b) Pointer-based Sorting

Figure 2: C code for two alternative versions of Bubble Sorting.

Figure 1 provides two alternative C versions for `LexiSort`. One version uses indexes to access the positions of the vector `S`, while the other version is based on pointers. You are free to use either of these versions as a basis for your RISC-V implementation of `LexiSort`. You may also create your own implementation of Bubble Sort as long as it correctly sorts the string pointers in the vector `S`. If using the pointer-based version, remember that the C compiler increments or decrements pointers based on their types. Thus, if a pointer `P` points to an item that is stored in `k` bytes, the C compiler will replace a statement such as `P++` by the statement `P=P+k`. Both versions of the C code above use a `break` statement inside a `for` loop. In the C language a `break` terminates the execution of the loop that contains it.

parameters:

- a0: Address of the first position of a vector S containing pointers to strings

return value: None

side effect: The pointers in S are now sorted according to the lexicographic order of the strings.

```
47 LexiSort:
48     addi    sp, sp, -32
49     sw      ra, 0(sp)
50     sw      s0, 4(s0)
51     sw      s1, 8(sp)
52     sw      s2, 12(sp)
53     sw      s3, 16(sp)
54     sw      s4, 20(sp)
55     sw      s5, 24(sp)
56     sw      s6, 28(sp)
57     mv      s0, a0          # s0 <- S
58     addi    s1, zero, -1    # i <- -1
59     addi    s6, zero, -1    # s6 <- -1
60 Loopi:
61     addi    s1, s1, 1        # i <- i+1
62     sll     t1, s1, 2        # t1 <- 4*i
63     add     t2, s0, t1       # t2 <- Address(S[i])
64     lw      t3, 0(t2)        # t3 <- S[i]
65     beq     t3, s6, Sorted   # if S[i] == -1 goto Sorted
66     addi    s2, s1, -1       # j <- i-1
67     blt     s2, zero, Loopi  # if j<0 goto Loopi
68 Loopj:
69     sll     t4, s2, 2        # t4 <- j*4
70     add     s3, s0, t4       # s3 <- Address(S[j])
71     lw      s4, 0(s3)        # Stempj <- S[j]
72     lw      s5, 4(s3)        # Stempj1 <- S[j+1]
73     mv      a0, s4           # a0 <- Stempj
74     mv      a1, s5           # s5 <- Stempj1
75     jal     LexiCmp          # Ctemp <- LexiCmp(Stempj,Stempj1)
76     ble     a0, zero, Loopi  # if Ctemp <= 0 goto Loopi
77     sw      s5, 0(s3)        # S[j] <- Stempj1
78     sw      s4, 4(s3)        # S[j+1] <- Stempj
79     addi    s2, s2, -1       # j <- j-1
80     bge     s2, zero, Loopj  # if j>=0 goto Loopj
81     j       Loopi           # goto Loopi
82 Sorted:
83     lw      ra, 0(sp)
84     lw      s0, 4(s0)
85     lw      s1, 8(sp)
86     lw      s2, 12(sp)
87     lw      s3, 16(sp)
88     lw      s4, 20(sp)
89     lw      s5, 24(sp)
90     lw      s6, 28(sp)
91     addi    sp, sp, 32
92     jalr    zero, ra, 0
```

Figure 3: An index-based solution to the LexiSort function.

```

114 LexiSort:
115     addi    sp, sp, -28
116     sw      ra, 0(sp)
117     sw      s0, 4(sp)
118     sw      s1, 8(sp)
119     sw      s2, 12(sp)
120     sw      s3, 16(sp)
121     sw      s4, 20(sp)
122     sw      s5, 24(sp)
123     mv      s0, a0          # s0 <- S
124     addi    s1, a0, -4      # T <- S-1
125     addi    s2, zero, -1    # s2 <- -1
126 Loopi:
127     addi    s1, s1, 4        # T++
128     lw      t1, 0(s1)        # t1 <- *T
129     beq     t1, s2, Sorted    # if *T == -1 goto sorted
130     addi    s3, s1, -4        # R <- T-1
131     blt     s3, s0, Loopi     # if R < S goto Loopi
132     lw      s4, 0(s3)        # Rtemp <- *R
133     lw      s5, 4(s3)        # Rtemp1 <- *(R+1)
134     mv      a0, s4           # a0 <- Rtemp
135     mv      a1, s5           # a1 <- Rtemp1
136     jal     Lexicmp          # Ctemp <- LexiCmp(Rtemp, Rtemp1)
137     ble     a0, zero, Loopi   # if Ctemp <= 0 goto Loopi
138     sw      s5, 0(s3)        # *R <- Rtemp1
139     sw      s4, 4(s3)        # *(R-1) <- Rtemp
140     addi    s3, s3, -4        # R--
141     bge     s2, s0, Loopj     # if R >= S goto Loopj
142     j       Loopi           # goto Loopi
143 Sorted:
144     lw      ra, 0(sp)
145     lw      s0, 4(sp)
146     lw      s1, 8(sp)
147     lw      s2, 12(sp)
148     lw      s3, 16(sp)
149     lw      s4, 20(sp)
150     lw      s5, 24(sp)
151     addi    sp, sp, 28
152     jalr    zero, ra, 0

```

Figure 4: A pointer-based solution to the LexiSort function.