Question 3 (13 points):

```
53 # Returns 1 if the input substring is a palindrome
54 # Returns 0 otherwise
55 # Arguments:
56 #     a0: Pointer to first character of the string
57 #     a1: position of first character of the substring
58 #     a2: position of the last character of the substring
59 #
60 palindrome:
61      bgt a2, a1 checkEnds # if last character > first character
62      li a0, 1             # otherwise it is done checking
63      ret
64 checkEnds:
65      add t0, a0, a1      # t0 <- Address of first character
66      lbu t1, 0(t0)       # t1 <- first character
67      add t2, a0, a2      # t2 <- Address of last character
68      lbu t3, 0(t2)       # t3 <- last character
69      beq t1, t3, recurse # if first character == last character
70      mv a0, zero         # first and last characters do not match
71      ret
72 recurse:
73      addi sp, sp, -4
74      sw ra, 0(sp)
75      addi a1, a1, 1      # move first character to the right
76      addi a2, a2, -1     # move last character to the left
77      jal ra, palindrome
78      lw ra, 0(sp)
79      addi sp, sp, 4
80      ret
```

Figure 1: RISC-V code for function `palindrome`

Figure 1 has the RISC-V assembly code for the recursive function `palindrome`. A string is a palindrome if it reads the same backward or forward. Examples: `deed`, `rotator`, `noon`. Assume that `SP = 0x04A00000` when this function is called with the following parameters:

- Register `a0` contains the address of the null terminated string "detected".
- `a1= 0x00000000`
- `a2 = 0x00000007`

a. (**5 points**) When this recursive function executes, it changes the value of the stack pointer register `SP`. What will be the lowest value written to the register `SP`, expressed in hexadecimal, while executing this recursive `palindrome` with the parameters above?

word: `detected`

- First call: first character: `d`; last character: `d`; `beq` is true, branch to `recurse`: `SP<-initialSP-4` and call again.
- Second call: first character: `e`; last character: `e`; `beq` is true, branch to `recurse`: `SP<-intialSP-8` and call again.
- Third call: first character: `t`; last character: `t`; `beq` is true, branch to `recurse`: `SP<-intialSP-12` and call again.

- Fourth call: first character: `e`; last character: `c`; `beq` is false, return.

Thus, the lowest value of `iniitalSP-12 = 0x04A0000 - 0xC = 0x049FFFF4`

b. (**3 points**) Assume that the instruction `bgt` in line 61 is at address `0x00400000`, what is the address, expressed in hexadecimal, where the instruction `beq` at line 69 is stored in memory?

The PC must advance 7 instructions to reach the `beq` instruction. 7*4 = 28 = 0x1C. Thus the `beq` instruction is at address `0x0040001C`,

c. (**5 points**) What is the binary representation, expressed in hexadecimal, of the instruction `beq` in line 69?

The format for a `beq` instruction is `SB` (from the green sheet).



`rs1 = t1 = x6 = 00110`, `rs2 = t3 = x28 = 11100`. The opcode for `beq` is `1100011` and the function code is `000`.

The immediate value to be added to the PC is 0x0000000C (3 instructions = 12 bytes). Thus: imm[3:2] = 11, all other bits of the immediate are 0.

Binary representation: `0000000 11100 00110 000 01100 1100011`
Binary representation: `0000 0001 1100 0011 0000 0110 0110 0011`
Hexadecimal: `0x01C3 0663`