

Question 5 (30 points):

Write the RISC-V assembly code for function `reverseAll`. The parameters for `reverseAll` are as follows:

- `a0`: the address of the first position of an array `S` of pointers to strings
- `a1`: the address of the first position of an array `L` of non-negative integers. For any index `i`, `L[i]` contains the length of the string whose address is in `S[i]`. The length of a string does not include the null character that terminates the string.

A position of the array `S` with the value `0xFFFFFFFF` is a sentinel that indicates the end of the array.

For each string whose address is in the array `S`, `reverseAll` must invoke the function `reverseString` from question 5 to reverse the order of the characters in the string.

`reverseAll` does not have any return values.

Your RISC-V code must follow all the register saving/restoring convention of RISC-V.

```

1  # reverseAll
2  # a0: address of first position of an array S.
3  #   Each element in S contains a pointer to a string.
4  # a1: address of first position of an array L. Each element L[i]
5  #   contains the length of the string whose pointer is in S[i].
6  # Pseudo code:
7  # ps <- address of first string pointer
8  # pl <- address of first length
9  # while(*ps != 0xFFFFFFFF)
10 #   reverseString(*ps, *pl)
11 #   ps++
12 #   pl++
13 reverseAll:
14     addi sp, sp, -16
15     sw s0, 0(sp)
16     sw s1, 4(sp)
17     sw s2, 8(sp)
18     sw ra, 12(sp)
19     mv s0, a0          # s0 <- initial string pointer
20     mv s1, a1          # s1 <- initial length pointer
21     li s2, -1          # s2 <- 0xFFFFFFFF
22     lw a0, 0(s0)        # a0 <- *ps
23     beq a0, s2, doneAll # if sentinel we are done
24 nextString:
25     lw a1, 0(s1)        # a1 <- *pl
26     jal reverseString   # reverseString
27     addi s0, s0, 4      # ps++
28     addi s1, s1, 4      # pl++
29     lw a0, 0(s0)        # a0 <- *ps
30     bne a0, s2, nextString # if not sentinel, go back
31 doneAll:
32     lw s0, 0(sp)
33     lw s1, 4(sp)
34     lw s2, 8(sp)
35     lw ra, 12(sp)
36     addi sp, sp, 16
37     jalr zero, ra, 0

```

Figure 1: A solution for reverseString.